線形周期変係数系の数値的安定判別法に関する汎用プログラムの開発

杉　岡　一　郎*
田　川　遼　三　郎*

(昭和43年11月26日受理)

Development of a General Program concerning a Numerical Method for Stability Analysis of Time-Varying Linear Systems with Periodic Coefficients

Ichiro SUGIOKA
Ryozaburo TAGAWA

(Received November 26, 1968)

Abstract

In various engineering fields, we often have a case in which the behavior of a system under consideration is described by a set of linear ordinary differential equations with periodic coefficients, and up to the present, various studies have been presented related to the stability analysis of such a system.

In this paper, a general program using especially the method of reference (1) is proposed, and also, the relation between the complexity of the system and the memory capacity of a computer were examined, and further, the relation between the complexity of the system and the computation time were examined.

1. 緒　　言

工学のいろいろな分野において，系の動作が線形周期変係数常微分方程式で記述される場合がしばしばあるが，このような系の安定判別に関してこれまで種々の研究[1)~4)]が発表されている。

特に本稿では，広範囲の線形周期変係数系の安定判別に適用できることを特徴とする文献(1)の方法に基づいて開発された電子計算機の汎用プログラムを示し，且つこのプログラムを使用する際に対象となる系の規模と計算機の記憶容量，ならびに計算時間との関係を考察している。

*　電気工学科　電気機械学第二講座

2. 汎用プログラムの構成

文献 (1) による安定判別の計算手順は次の通りである。

(1) 与えられた系の方程式を数値的に解くことにより transition matrix[5),6)] を求める。

(2) この transition matrix から系の特性方程式を誘導する。

(3) 更にこの特性方程式に例えば Schur-Cohn の安定判別法[7)] を適用して根の存在領域を判定し系の安定性を決定する。

汎用プログラムでは，主として汎用という立前からこれらの段階を次のように処理している。

まず (1) の段階では種々の数値解法が利用できるが，ここでは Runge-Kutta-Gill の方法[8),9)] を用いた。この方法は汎用の目的からばかりではなく精度の観点からも，また有利である。

(2) の段階では予め transition matrix の次数により表を作っておく方法，Frame 法[10)]，および Danilevskii 法[9),10)] を用いることができるが，ここでは記憶容量および精度の観点から Danilevskii 法を採用した。

(3) の段階では特性方程式の係数により作られる条件式 (系の安定性を保証する必要条件式) を系の次数に応じて作表しておく方法も考えられるが，ここでは最も汎用性にすぐれた手段として Schur-Cohn の安定判別法に従って忠実にプログラムを作成した。したがってこの段階は次の 3 つの部分から構成される。

(i) 特性方程式の係数から安定判別のための行列式を順番に作る部分。

(ii) (i) で作られた行列式の値を計算する部分。この部分では種々の数値計算法が利用できるが汎用性，精度の観点から掃出し法[8),10)] を採用した。

(iii) (i) で作られた行列式の順番とその行列式の値の符号との関係を検討して，系の安定性を決定する部分。

全体のプログラムはこれらの各段階を適当に組み合わせることによって Fig. 1 のように構成される。

Fig. 1. Block diagram of a general program.

したがってこのプログラムを用いるならば対象となる系が変更されても，

(a) 系の方程式とその次数表示

(b) 係数の判別領域およびそのきざみ幅指定

(c) 係数の周期表示

(d) 数値解析の時間原点指定

(e) 数値解析のステップ幅指定

(f) 数値解析のステップ数指定

(g) 安定化係数[1] 指定

を変更するのみで極めて広範囲の線形周期変係数系の安定判別が可能である。

3. 必要な計算機の記憶容量と計算時間

必要なメモリの数は対象とする系の次数を n, 判別結果の記憶に必要なメモリの数を A とすると, メインプログラムで約 $(5n^2+n+200+A)$ 語, (1) の段階で約 $(3n+400)$ 語, (2) の段階

Table 1. Computer program for example 1.

```
C     STABILITY ANALYSIS OF SYSTEM WITH DELAY ELEMENT
      DIMENSION A(11,11),XX(12),AB(22,22),INDEX(11,13)
      READ(5,111) KN,M,L1,L2
  111 FORMAT(4I12)
      READ(5,112) COEFO1,COEFO2,ALPHA1,ALPHA2,TI,TO,STEP,SF
  112 FORMAT(5F12.0)
      COEF2=COEFO2
      DO 10 I=1,L2
      COEF1=COEFO1
      DO 20 J=1,L1
      N=KN
      CALL TRAMAT(N,M,COEF1,COEF2,TI,TO,STEP,A)
      CALL DANILE(N,A,SF,XX,IND)
      IF(IND) 300,400,500
  400 K=1
  700 CALL SCHCOH(N,K,XX,AB)
      CALL DETERM(K,AB,DELTA)
      CALL HANTEI(K,DELTA,IND)
      GO TO (600,500),IND
  600 K=K+1
      IF(K-N) 700,700,300
  300 INDEX(I,J)=0
      GO TO 30
  500 INDEX(I,J)=1
   30 COEF1=COEF1+ALPHA1
   20 CONTINUE
      COEF2=COEF2+ALPHA2
   10 CONTINUE
      WRITE(6,100) ((INDEX(I,J),J=1,L1),I=1,L2)
  100 FORMAT(1H1,/////,(1H0,13I3))
      STOP
      END
```

```
      SUBROUTINE TRAMAT(N,M,COEF1,COEF2,TI,TO,H,A)
      DIMENSION X(11),Q(11),A(1,1)
      E1=0.5
      E2=(1.-SQRT(E1))
      E3=(1.+SQRT(E1))
      E4=1./6.
      DO 5 L=1,N
      DO 6 J=1,N
      X(J)=0.
    6 Q(J)=0.
      X(L)=1.
      T=TO
      DO 1000 K=1,M
      CALL KANSU(T,X,COEF1,COEF2,TI,F)
      DO 10 I=1,N
      AK=H*F(I)
      QI=Q(I)
      R=E1*AK-QI
      X(I)=X(I)+R
   10 Q(I)=QI+3.*R-E1*AK
      CALL KANSU(T,X,COEF1,COEF2,TI,F)
      DO 20 I=1,N
      AK=H*F(I)
      QI=Q(I)
      R=E2*(AK-QI)
      X(I)=X(I)+R
   20 Q(I)=QI+3.*R-E2*AK
      CALL KANSU(T,X,COEF1,COEF2,TI,F)
      DO 30 I=1,N
      AK=H*F(I)
      QI=Q(I)
      R=E3*(AK-QI)
      X(I)=X(I)+R
   30 Q(I)=QI+3.*R-E3*AK
      CALL KANSU(T,X,COEF1,COEF2,TI,F)
      DO 40 I=1,N
      AK=H*F(I)
      QI=Q(I)
      R=E4*(AK-2.*QI)
      X(I)=X(I)+R
   40 Q(I)=QI+3.*R-E1*AK
      T=T0+H*FLOAT(K)
 1000 CONTINUE
      DO 50 I=1,N
   50 A(I,L)=X(I)
    5 CONTINUE
      RETURN
      END


      SUBROUTINE KANSU(T,X,COEF1,COEF2,TI,F)
      DIMENSION X(1),F(1)
      DO 10 I=1,10
   10 F(I)=20.*(X(I+1)-X(I))
      F(11)=-(X(11)+(COEF1+COEF2*SIN(6.283185/TI*T))*X(1))
      RETURN
      END
```

で約 $(9n^2+4n+800)$ 語，(3) の段階で約 $(3n^2+8n+700)$ 語である。したがってプログラム全体で必要なメモリの数は大体 $(17n^2+16n+2100+A)$ 語であると目安をつけることができる。

次に計算時間であるが，必要な計算時間の大部分は transition matrix を求める段階で費やされるものと考えられる。transition matrix を求める際系の次数が m 倍になれば常微分方程式を解く時の計算量が大体 m 倍になり，更に初期条件の組も m 倍になるので計算時間は結局 m^2 倍になるものと目安をつけることができる。

4. 計算例およびプログラム

このプログラムを使用して安定判別を行なった例を示す。計算機は東大大型計算機センターの HITAC5020 を使用した。

例 1. Fig. 2 のようなむだ時間要素を含む 1 次の系[1)] の安定領域を決定する。

むだ時間要素は (1) 式のような極限で表わすことができる。

$$e^{-sL}=\lim_{n\to\infty}\left(\frac{1}{\dfrac{L}{n}S+1}\right)^n \tag{1}$$

```
      SUBROUTINE DANILE(N,A,SF,XX,IND)
      DIMENSION A(1,1),AA(11),BB(11),C(11),X(12),XX(1),T(11,11),
     1TI(11,11),E(11,11),R(11,11),P(11,11),AB(22,22)
      Q=1.E-20
    3 K=1
      IF(N-(K+1)) 240,65,65
   65 AMAX=ABS(A(K+1,K))
      II=K+1
      IK=K+2
      IF(IK.GT.N) GO TO 500
      DO 15 I=IK,N
      IF(AMAX-ABS(A(I,K))) 16,15,15
   16 AMAX=ABS(A(I,K))
      II=I
   15 CONTINUE
  500 IF(AMAX.LE.Q) GO TO 240
      IF(II.EQ.(K+1)) GO TO 10
      DO 230 KK=1,N
      AA(KK)=A(K+1,KK)
      A(K+1,KK)=A(II,KK)
  230 A(II,KK)=AA(KK)
      DO 250 LL=1,N
      BB(LL)=A(LL,K+1)
      A(LL,K+1)=A(LL,II)
  250 A(LL,II)=BB(LL)
      GO TO 10
  240 DO 260 I=1,K
      IS=K-I+1
  260 C(I)=A(IS,K)
      DO 261 I=1,K+1
      IF(I.GE.(K+1)) GO TO 262
      IR=K-I+1
      X(I)=-C(IR)
      GO TO 265
  262 X(I)=1.
  265 XX(I)=X(I)
      IF(I.EQ.1) GO TO 261
      XX(I)=XX(I)*SF**(I-1)
  261 CONTINUE
      KH=1
  330 CALL SCHCOH(K,KH,XX,AB)
      CALL DETERM(KH,AB,DELTA)
      CALL HANTEI(KH,DELTA,IND)
      GO TO (310,320),IND
  310 KH=KH+1
      IF(KH.LE.K) GO TO 330
      N=N-K
      IF(N-1) 319,350,350
  350 DO 270 I=1,N
      DO 270 J=1,N
      IK=I+K
      JK=J+K
  270 A(I,J)=A(IK,JK)
      GO TO 3
   10 CALL UNIMAT(N,E)
      DO 20 I=1,N
      DO 20 J=1,N
      IF(J.EQ.(K+1)) GO TO 40
      T(I,J)=E(I,J)
      GO TO 20
   40 T(I,J)=A(I,K)
   20 CONTINUE
      DO 50 I=1,N
      DO 50 J=1,N
      IF(J-(K+1)) 60,35,60
   35 IF(I-(K+1)) 45,70,45
   45 TI(I,J)=-A(I,K)/A(K+1,K)
      GO TO 50
   70 TI(I,J)=1./A(K+1,K)
      GO TO 50
   60 TI(I,J)=E(I,J)
   50 CONTINUE
      DO 110 I=1,N
      DO 110 J=1,N
      R(I,J)=0.
      DO 110 L=1,N
  110 R(I,J)=R(I,J)+TI(I,L)*A(L,J)
      DO 120 I=1,N
      DO 120 J=1,N
      P(I,J)=0.
      DO 120 L=1,N
      P(I,J)=P(I,J)+R(I,L)*T(L,J)
  120 A(I,J)=P(I,J)
      K=K+1
      IF(K.LE.(N-1)) GO TO 65
      DO 130 I=1,N
      IT=N-I+1
  130 C(I)=A(IT,N)
      DO 140 I=1,N+1
      IF(I.GE.(N+1)) GO TO 150
      IU=N-I+1
      X(I)=-C(IU)
      GO TO 141
  150 X(I)=1.
  141 XX(I)=X(I)
      IF(I.EQ.1) GO TO 140
      XX(I)=XX(I)*SF**(I-1)
  140 CONTINUE
      IND=0
      GO TO 320
  319 IND=-1
  320 RETURN
      END
```

したがってここでは $n=10$ としてむだ時間要素を (2) 式で近似すると，Fig. 2 の系は Fig. 3 の系と近似的に等価である。

$$e^{-0.5s} = \left(\frac{20}{S+20}\right)^{10} \tag{2}$$

Table 1 は Fig. 3 の系について係数 k_1, k_2 を変化して安定領域を決定するプログラムであり，Table 2 は決定された安定領域である。

必要なメモリの数は約 4500 語，計算時間は約 12 分である。

Table 1 において，

SUBROUTINE TRAMAT : transition matrix の誘導

SUBROUTINE KANSU : 系の方程式の定義

SUBROUTINE DANILE : transition matrix から系の特性方程式の係数の誘導

SUBROUTINE UNIMAT : 単位行列の作成

SUBROUTINE SCHCOH : Schur-Cohn の安定判別法のための行列式の作成

SUBROUTINE DETERM : 行列式の値の計算

```
      SUBROUTINE UNIMAT(N,E)
      DIMENSION E(1,1)
      DO 10 I=1,N
      DO 10 J=1,N
      IF(I.EQ.J) GO TO 20
      E(I,J)=0.
      GO TO 10
   20 E(I,J)=1.
   10 CONTINUE
      RETURN
      END

      SUBROUTINE SCHCOH(N,KK,XX,AB)
      DIMENSION XX(1),AB(1,1),BA(11,11),AC(11,11),CA(11,11)
      DO 10 I=1,KK
      DO 10 J=1,KK
      IF(I.GE.J) GO TO 30
      AB(I,J)=0.
      GO TO 10
   30 IJ=I-J+1
      AB(I,J)=XX(IJ)
   10 CONTINUE
      DO 110 I=1,KK
      DO 110 J=1,KK
  110 BA(J,I)=AB(I,J)
      DO 20 I=1,KK
      DO 20 J=1,KK
      IF(I.GE.J) GO TO 40
      AC(I,J)=0.
      GO TO 20
   40 IN=N-I+J+1
      AC(I,J)=XX(IN)
   20 CONTINUE
      DO 120 I=1,KK
      DO 120 J=1,KK
  120 CA(J,I)=AC(I,J)
      DO 50 I=1,KK
      DO 50 J=1,KK
      JKK=J+KK
   50 AB(I,JKK)=CA(I,J)
      DO 60 I=1,KK
      DO 60 J=1,KK
      IKK=I+KK
      JKK=J+KK
      AB(IKK,J)=AC(I,J)
   60 AB(IKK,JKK)=BA(I,J)
      RETURN
      END
```

```
      SUBROUTINE DETERM(N,AB,DELTA)
      DIMENSION AB(1,1),AA(22),BB(22),RATIO(22),PIVOT(22)
      IR=0
      IC=0
      Q=1.E-20
      NN=N*2
      DO 100 K=1,NN
      II=K
      JJ=K
      AMAX=ABS(AB(K,K))
      DO 10 I=K,NN
      DO 10 J=K,NN
      IF(AMAX.GE.ABS(AB(I,J))) GO TO 10
      AMAX=ABS(AB(I,J))
      II=I
      JJ=J
   10 CONTINUE
      IF(AMAX.LE.Q) GO TO 90
      IF(II.EQ.K) GO TO 30
      DO 20 I=1,NN
      AA(I)=AB(K,I)
      AB(K,I)=AB(II,I)
   20 AB(II,I)=AA(I)
      IR=IR+1
   30 IF(JJ.EQ.K) GO TO 50
      DO 40 J=1,NN
      BB(J)=AB(J,K)
      AB(J,K)=AB(J,JJ)
   40 AB(J,JJ)=BB(J)
      IC=IC+1
   50 PIVOT(K)=AB(K,K)
      IF(K.GE.NN) GO TO 100
      DO 60 I=1,NN
      RATIO(I)=AB(K,I)/PIVOT(K)
   60 AB(K,I)=RATIO(I)
      DO 99 I=1,NN
      IF(I.EQ.K) GO TO 99
      Y=AB(I,K)
      DO 70 J=1,NN
      AB(I,J)=AB(I,J)-Y*RATIO(J)
   70 CONTINUE
   99 CONTINUE
  100 CONTINUE
      DELTA=1.
      DO 80 L=1,NN
   80 DELTA=PIVOT(L)*DELTA
      DELTA=(-1.)**(IR+IC)*DELTA
      GO TO 500
   90 DELTA=0.
  500 RETURN
      END
```

```
      SUBROUTINE HANTEI(K,DELTA,IND)
      L=K/2*2
      IF(K.GT.L) GO TO 20
      IF(DELTA) 15,15,30
   20 IF(DELTA) 30,15,15
   30 IND=1.
      GO TO 100
   15 IND=2
  100 RETURN
      END
```

```
         11        100        13        11
5.        10.       1.        -1.        1.
0.        0.01      1.
```

Table 2. The results of stability analysis for the system expressed in Fig. 2. (SF=1)

```
      10-*  *  0  0  0  *  *  *  *  *  *  *  *
        .*  *  *  0  0  *  *  *  *  *  *  *  *
     · 8-*  *  *  0  0  0  *  *  *  *  *  *  *
     ·   *  *  *  *  0  0  *  *  *  *  *  *  *
       6-*  *  *  *  0  0  0  *  *  *  *  *  *
 k2      *  *  *  *  0  0  0  *  *  *  *  *  *
       4-*  *  *  *  0  0  0  0  *  *  *  *  *
         *  *  *  *  0  0  0  0  *  *  *  *  *
       2-*  *  *  *  0  0  0  0  0  0  *  *  *
         *  *  *  *  0  0  0  0  0  0  *  *  *
       0-*  *  *  *  *  0  0  0  0  0  *  *  *
         '     '     '  '  '     '     '     '
        -5    -3    -1  0  1     3     5     7
                        k1       { 0:stable
                                 { *:unstable
```

SUBROUTINE HANTEI: 系の安定性の判定

のための subroutine である。



Fig. 2. First order system with a delay element for first example.



Fig. 3. Block diagram equivalent to that in Fig. 2.

例 2. Fig. 4 に示す 3 次の系[1),2)] ($\omega_0=2$ rad/sec, $\omega_0=10$ rad/sec) の安定領域を決定する。Table 3 および Table 4 は決定された安定領域である。

必要なメモリの数は約 2800 語，計算時間は約 4 分である。

例 3[1)]. 次に (3) 式で示される Mathieu の微分方程式の安定領域[11)]を決定する。



Fig. 4. Third order system for second example.



Fig. 5. Block diagram equivalent to that in Fig. 4.

$$\frac{d^2x}{dt^2}+(k_1+k_2\cos t)\,x=0 \tag{3}$$

Table 5 は決定された安定領域である。

必要なメモリの数は約 3600 語，計算時間は約 12 分である。

Table 3. The results of stability analysis for the system expressed in Fig. 4, $\omega_0=2$ rad/sec. (SF=1)

10-	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*
	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*
9-	*	*	*	*	*	*	*	0	*	*	*	*	*	*	*	*	*	*	*	*	*
	*	*	0	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*
8-	*	*	0	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*
	*	*	0	*	*	*	*	*	0	*	*	*	*	*	*	*	*	*	*	*	*
7-	*	*	0	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*
	*	*	*	*	*	*	*	*	*	0	*	*	*	*	*	*	*	*	*	*	*
6-	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*
	*	*	*	*	*	*	*	*	*	*	0	*	*	*	*	*	*	*	*	*	*
k_2 5-	*	*	*	0	*	*	*	*	*	*	*	0	*	*	*	*	*	*	*	*	*
	*	*	*	0	*	*	*	*	*	*	*	*	0	*	*	*	*	*	*	*	*
4-	*	*	*	0	*	*	*	*	*	*	*	*	*	0	*	*	*	*	*	*	*
	*	*	*	*	0	*	*	*	*	*	*	*	*	0	0	*	*	*	*	*	*
3-	*	*	*	*	0	*	*	*	*	*	*	*	*	*	0	0	0	*	*	*	*
	*	*	*	*	0	0	*	*	*	*	*	*	*	*	*	0	0	0	0	*	*
2-	*	*	*	*	0	0	0	*	*	*	*	*	*	*	*	0	0	0	0	0	0
	*	*	*	*	*	0	0	0	0	*	*	*	*	*	0	0	0	0	0	0	0
1-	*	*	*	*	*	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
	*	*	*	*	*	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
0-	*	*	*	*	*	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
	0					5					10					15					20

k_1

{ 0:stable
{ *:unstable

Table 4. The results of stability analysis for the system expressed in Fig. 4, $\omega_0=10$ rad/sec. (SF=1)

40-	*	*	0	0	0	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*
	*	*	0	0	0	0	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*
	*	*	0	0	0	0	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*
	*	*	0	0	0	0	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*
	*	*	0	0	0	0	*	*	*	*	*	*	*	*	*	*	*	*	*	*	*
30-	*	*	0	0	0	0	0	*	*	*	*	*	*	*	*	*	*	*	*	*	*
	*	*	0	0	0	0	0	*	*	*	*	*	*	*	*	*	*	*	*	*	*
	*	*	0	0	0	0	0	0	*	*	*	*	*	*	*	*	*	*	*	*	*
	*	*	*	0	0	0	0	0	*	*	*	*	*	*	*	*	*	*	*	*	*
	*	*	*	0	0	0	0	0	0	*	*	*	*	*	*	*	*	*	*	*	*
k_2 20-	*	*	*	0	0	0	0	0	0	0	*	*	*	*	*	*	*	*	*	*	*
	*	*	*	0	0	0	0	0	0	0	0	*	*	*	*	*	*	*	*	*	*
	*	*	*	0	0	0	0	0	0	0	0	0	*	*	*	*	*	*	*	*	*
	*	*	*	0	0	0	0	0	0	0	0	0	0	*	*	*	*	*	*	*	*
	*	*	*	*	0	0	0	0	0	0	0	0	0	0	0	*	*	*	*	*	*
10-	*	*	*	*	0	0	0	0	0	0	0	0	0	0	0	0	0	0	*	*	*
	*	*	*	*	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
	*	*	*	*	*	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
	*	*	*	*	*	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
	*	*	*	*	*	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
0-	*	*	*	*	*	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
	0					5					10					15					20

k_1

{ 0:stable
{ *:unstable

Table 1 のプログラムで使われている主な変数名の説明

M: 数値解析におけるステップ数
L1: 係数 k_1 の個数
L2: 係数 k_2 の個数
COEFO1: 係数 k_1 の初期値
COEFO2: 係数 k_2 の初期値
COEF1: 係数 k_1
COEF2: 係数 k_2
ALPHA1: 係数 k_1 のきざみ幅
ALPHA2: 係数 k_2 のきざみ幅
TI: 係数の周期
TO: 数値解析における時間原点
SF: 安定化係数
A(i, j): transition matrix
XX(i): 特性方程式の係数
AB(i, j): Schur-Cohn の安定判別法のための行列式
IND: 安定か不安定か更に計算を続けるか否かの判定結果
DELTA: 行列式の値
T: 数値解析における時刻
X(i): x_i
F(i): $\dot{x}_i$
E(i, j): 単位行列

—— 以上の変数名はプログラム全体に共通 ——

MAIN PROGRAM.
KN: 系の次数
STEP: 数値解析におけるステップ幅
INDEX(i, j): 係数 k_{1_i}, k_{2_j} に対する判別結果

SUBROUTINE TRAMAT.
H: 数値解析におけるステップ幅
N: 常微分方程式の階数

SUBROUTINE DANILE.
N: transition matrix の次数

SUBROUTINE UNIMAT.
N: 単位行列の次数

SUBROUTINE SCHCOH.
N: 特性方程式の次数
KK: Schur-Cohn の安定判別法のための行列式の順番

SUBROUTINE DETERM.
N: Schur-Cohn の安定判別法のための行列式の順番
NN: 行列式の次数
PIVOT(i): 掃出し法における枢軸

SUBROUTINE HANTEI.
K: Schur-Cohn の安定判別法のための行列式の順番

Table 5. The results of stability analysis for Equ. (3). (SF=1.002)

```
3.5-* * * * * * * * * * * * * * * * * * * * * 0 0 * * * * * * * * * * * 0 0 0 0 0 0
    * * * * * * * * * * * * * * * * * * * * * 0 0 * * * * * * * * * * 0 0 0 0 0 0 0
    * * 0 * * * * * * * * * * * * * * * * * * 0 0 * * * * * * * * * * 0 0 0 0 0 0 0
    * * * * * * * * * * * * * * * * * * * * 0 0 0 * * * * * * * * * * 0 0 0 0 0 0 0
    * * * * * * * * * * * * * * * * * * * * 0 0 0 * * * * * * * * * 0 0 0 0 0 0 0 0
3. -* * * * * * * * * * * * * * * * * * * * 0 0 0 * * * * * * * * * 0 0 0 0 0 0 0 0
    * * * * * * * * * * * * * * * * * * * * 0 0 0 * * * * * * * * 0 0 0 0 0 0 0 0 0
    * * * * 0 * * * * * * * * * * * * * * * 0 0 0 * * * * * * * * 0 0 0 0 0 0 0 0 0
    * * * * * * * * * * * * * * * * * * * * 0 0 0 0 * * * * * * 0 0 0 0 0 0 0 0 0 0
    * * * * * * * * * * * * * * * * * * * 0 0 0 0 0 * * * * * * 0 0 0 0 0 0 0 0 0 0
2.5-* * * * * 0 * * * * * * * * * * * * * 0 0 0 0 0 * * * * * 0 0 0 0 0 0 0 0 0 0 0
    * * * * * 0 * * * * * * * * * * * * * 0 0 0 0 0 * * * * * 0 0 0 0 0 0 0 0 0 0 0
    * * * * * * * * * * * * * * * * * * * 0 0 0 0 0 * * * * 0 0 0 0 0 0 0 0 0 0 0 0
    * * * * * * 0 * * * * * * * * * * * 0 0 0 0 0 0 * * * * 0 0 0 0 0 0 0 0 0 0 0 0
    * * * * * * 0 * * * * * * * * * * * 0 0 0 0 0 0 * * * * 0 0 0 0 0 0 0 0 0 0 0 0
2. -* * * * * * 0 * * * * * * * * * * * 0 0 0 0 0 0 * * * 0 0 0 0 0 0 0 0 0 0 0 0 0
    * * * * * * 0 0 * * * * * * * * * 0 0 0 0 0 0 0 * * * 0 0 0 0 0 0 0 0 0 0 0 0 0
k2  * * * * * * * 0 * * * * * * * * * 0 0 0 0 0 0 0 * * 0 0 0 0 0 0 0 0 0 0 0 0 0 0
    * * * * * * * 0 * * * * * * * * 0 0 0 0 0 0 0 0 * * 0 0 0 0 0 0 0 0 0 0 0 0 0 0
    * * * * * * * 0 * * * * * * * * 0 0 0 0 0 0 0 0 * * 0 0 0 0 0 0 0 0 0 0 0 0 0 0
1.5-* * * * * * * 0 0 * * * * * * * 0 0 0 0 0 0 0 0 * 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
    * * * * * * * 0 0 * * * * * * 0 0 0 0 0 0 0 0 0 * 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
    * * * * * * * 0 0 * * * * * * 0 0 0 0 0 0 0 0 * * 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
    * * * * * * * 0 0 * * * * * 0 0 0 0 0 0 0 0 0 * 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
    * * * * * * * 0 0 0 * * * * 0 0 0 0 0 0 0 0 0 * 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
1. -* * * * * * 0 0 0 0 * * * 0 0 0 0 0 0 0 0 0 0 * 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
    * * * * * * 0 0 0 0 * * * 0 0 0 0 0 0 0 0 0 0 * 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
    * * * * * * 0 0 0 0 * * * 0 0 0 0 0 0 0 0 0 0 * 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
    * * * * * * 0 0 0 0 * * 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
    * * * * * 0 0 0 0 0 * * 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0.5-* * * * * 0 0 0 0 0 * 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
    0 * * * * 0 0 0 0 0 * 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
    0 * * * 0 0 0 0 0 0 * 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
    0 0 * * 0 0 0 0 0 0 * 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
    0 0 * 0 0 0 0 0 0 0 * 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0. -0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
    |         |         |         |         |         |         |         |
    0.        0.5       1.        1.5       2.        2.5       3.        3.5

                                  k1                        0:stable
                                                            *:unstable
```

5. 結　　言

以上文献 (1) の方法に基づいて開発された線形周期変係数系の安定判別用の汎用プログラムを示し，且つこのプログラムを使用する際に計算機がどの程度の記憶容量を必要とするかについて検討し，対象とする系の規模と計算機の記憶容量および計算時間の関係などについて述べた。また例 (1)～(3) によってこれらのことを確かめた。

特に高い精度を要する場合には各段階の計算精度の評価が重要な問題となるが，これについては今後更に研究を進める予定である。

終りに日頃御討論いただく土谷武士助教授ならびに計算機の使用に際しお世話になった北大計算センターの相良砌氏をはじめ職員の皆様に深謝申し上げます。

参 考 文 献

1) R. Tagawa: A numerical method for stability analysis of time-varying linear systems with periodic coefficients, Memoirs of Faculty of Engineering, Hokkaido University, Vol. XII, No. 1 (No. 54), January, 1967.

2) W. W. Cooley, R. N. Clark, R. C. Buckner: Stability in linear systems having a time-variable parameter, IEEE Trans., Vol. AC-9, No. 4, October, 1964.

3) R. Tagawa, R. Miura: A method of analysis for feed-back systems containing a transfer element with a gain constant varying periodically, Memoirs of Faculty of Engineering, Hokkaido University, Vol. XI, No. 4, March, 1963.

4) 江尻: 周期変動ダンピング2次系の安定性, 計測自動制御学会論文集, 第2巻, 第2号 (昭41).

5) J. T. Tou: Modern control theory, McGraw-Hill, 1964.

6) L. A. Zadeh, C. A. Desoer: Linear system theory, McGraw-Hill 1963.

7) E. I. Jury: Sampled data control systems, John Wiley & Sons, 1958.

8) 山内・森口・一松: 電子計算機のための数値計算法 I (昭40), 培風館.

9) 宇野: 計算機のための数値計算 (昭38), 朝倉書店.

10) 情報処理学会: 電子計算機ハンドブック (昭41), オーム社.

11) 三木: 常微分方程式とその応用 (昭41), コロナ社.